Assignment 2 - solutions

Jacques Carette

```
> restart;
> with(VectorCalculus):
```

Q1

The position, in cartesian coordinates is given by a piecewise function; the pieces are (as vectors):

```
> p1,p2 := PositionVector([t,t^3,0]),
  PositionVector([t,0,t^3]);
```

$$p1, p2 := \begin{bmatrix} t \\ t^3 \\ 0 \end{bmatrix}, \begin{bmatrix} t \\ 0 \\ t^3 \end{bmatrix}$$

We need to show that the position, velocity and acceleration are continuous, but that the Normal vector is not.

```
> r := zip((a,b)->piecewise(t<0,a,b),p1,p2);
```

$$r := \begin{bmatrix} \begin{cases} t & t < 0 \\ t & \mathit{otherwise} \end{cases} \\ \begin{cases} t^3 & t < 0 \\ 0 & \mathit{otherwise} \end{cases} \\ \begin{cases} 0 & t < 0 \\ t^3 & \mathit{otherwise} \end{cases} \end{bmatrix}$$

We need to compute the limit at 0 of the position (internally computes left and right limits):

```
> map(limit,r,t=0);
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

The velocity can then be derived as

```
> v := diff(r,[t]);
```

$$v := \begin{bmatrix} 1 \\ \begin{cases} 3\,t^2 & t \le 0 \\ 0 & 0 < t \end{cases} \\ \begin{cases} 0 & t \le 0 \\ 3\,t^2 & 0 < t \end{cases} \end{bmatrix}$$

We can do the same thing for the velocity

```
> map(limit, v, t=0);
```

$$\begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix}$$

and the acceleration too:

```
> a := diff(v, t):  map(limit, a, t=0);
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

The simplest way to define a normal is to find a unit vector which is perpendicular to T(t). Here we will save ourselves some trouble:

```
> N := map(simplify, TNBFrame(r, t, output=['N'])) assuming t::real;
```

$$N := \begin{bmatrix} -\frac{3|t|t}{\sqrt{1+9t^4}} \\ \begin{cases} -\frac{1}{\sqrt{1+9t^4}} & t<0 \\ 1 & t=0 \\ 0 & 0<t \end{cases} \\ \begin{cases} 0 & t\le 0 \\ \frac{1}{\sqrt{1+9t^4}} & 0<t \end{cases} \end{bmatrix}$$

And now we can compute the limits:

```
> map(limit, N, t=0, left), map(limit, N, t=0, right);
```

$$\begin{bmatrix} 0 \\ -1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix}$$

from which we can see that the normal moves discontinuously

```
>
```

Q2

Let a curve C in cartesian coordinates be given by

```
> r := PositionVector([(sin(t)*cos(t)*(16*cos(t)^4-20*cos(t)^2+5), (16*cos(t)^4-12*cos(t)^2+1)*sin(t)^2, (16*cos(t)^4-12*cos(t)^2+1)*sin(t)^2)]);
```

$$r := \begin{bmatrix} \sin(t)\cos(t)\left(16\cos(t)^4-20\cos(t)^2+5\right) \\ \left(16\cos(t)^4-12\cos(t)^2+1\right)\sin(t)^2 \\ \left(16\cos(t)^4-12\cos(t)^2+1\right)\sin(t)^2 \end{bmatrix}$$

As the hint says, this can be simplified since the above contains an expanded form of cos(5*t) and sin(5*y), viz

```
> expand(cos(5*t)), expand(sin(5*t));
```

$$16\cos(t)^5-20\cos(t)^3+5\cos(t),\ 16\sin(t)\cos(t)^4-12\sin(t)\cos(t)^2+\sin(t)$$

So we can see that r has a common factor of sin(t), and the rest being:

```
> r1 := map(combine, expand(r/sin(t)));
```

$$r1 := \begin{bmatrix} \cos(5t) \\ \sin(5t) \\ \sin(5t) \end{bmatrix}$$

thus we really need to work with

```
> r2 := map(x->x*sin(t),r1);
```

$$r2 := \begin{bmatrix} \cos(5t)\sin(t) \\ \sin(5t)\sin(t) \\ \sin(5t)\sin(t) \end{bmatrix}$$

We can change coordinates:

```
> Cspherical := MapToBasis(Vector(r2), spherical);
```

$$Cspherical := \sqrt{\cos(5t)^2\sin(t)^2 + 2\sin(5t)^2\sin(t)^2}\,\mathbf{e}_{\mathbf{r}} + \arctan\left(\sqrt{\cos(5t)^2\sin(t)^2 + \sin(5t)^2\sin(t)^2}, \sin(5t)\sin(t)\right)\mathbf{e}_{\phi} + \arctan(\sin(5t)\sin(t), \cos(5t)\sin(t))\,\mathbf{e}_{\theta}$$

This is messy. However, one can fairly easily see that the expression for theta is arctan(tan(5*t)), so that theta=5*t. Phi comes out to arctan(|sin(t)|/(sin(t)*sin(5*t)). Since arctan is even, that is arctan(1/sin(5*t)). r comes out to |sin(t)| * sqrt(1+sin(5*t)^2).

The main conclusion is that there was a transcription error somewhere... I meant theta=5*t, phi = t, and r=1 ! I am surprised no one caught this.

We can still compute the answer, but it is messy

```
> v := diff(Cspherical, t);
```

$$v := \frac{1}{2}\frac{10\cos(5t)\sin(t)^2\sin(5t) + 2\cos(5t)^2\sin(t)\cos(t) + 4\sin(5t)^2\sin(t)\cos(t)}{\sqrt{\cos(5t)^2\sin(t)^2 + 2\sin(5t)^2\sin(t)^2}}\mathbf{e}_{\mathbf{r}} + \left(\frac{1}{2}\frac{2\cos(5t)^2\sin(t)\cos(t) + 2\sin(5t)^2\sin(t)\cos(t)}{\sqrt{\cos(5t)^2\sin(t)^2 + \sin(5t)^2\sin(t)^2}\,\sin(5t)\sin(t)} - \frac{5\sqrt{\cos(5t)^2\sin(t)^2 + \sin(5t)^2\sin(t)^2}\,\cos(5t)}{\sin(5t)^2\sin(t)} - \frac{\sqrt{\cos(5t)^2\sin(t)^2 + \sin(5t)^2\sin(t)^2}\,\cos(t)}{\sin(5t)\sin(t)^2}\right) \Bigg/ \left(1 + \frac{\cos(5t)^2\sin(t)^2 + \sin(5t)^2\sin(t)^2}{\sin(5t)^2\sin(t)^2}\right)\mathbf{e}_{\phi} + \frac{5 + \frac{5\sin(5t)^2}{\cos(5t)^2}}{1 + \frac{\sin(5t)^2}{\cos(5t)^2}}\mathbf{e}_{\theta}$$

```
> v2 := combine(simplify(v));
```

$$v2 := \frac{-3\sin(12t) - 2\sin(8t) + 5\sin(10t) + 3\sin(2t)}{\sqrt{12 - 4\cos(10t) - 12\cos(2t) + 2\cos(8t) + 2\cos(12t)}}\mathbf{e}_{\mathbf{r}} + \frac{10\,\mathrm{csgn}(\sin(t))\cos(5t)}{-3 + \cos(10t)}\mathbf{e}_{\phi} + 5\,\mathbf{e}_{\theta}$$

```
> a := diff(v2, t);
```

$$a := \left(\frac{-36\cos(12t) - 16\cos(8t) + 50\cos(10t) + 6\cos(2t)}{\sqrt{12 - 4\cos(10t) - 12\cos(2t) + 2\cos(8t) + 2\cos(12t)}} - \frac{1}{2} (-3\sin(12t) - 2\sin(8t) + 5\sin(10t) + 3\sin(2t)) (40\sin(10t) + 24\sin(2t) - 16\sin(8t) - 24\sin(12t)) / (12 - 4\cos(10t) - 12\cos(2t) + 2\cos(8t) + 2\cos(12t))^{(3/2)} \right) \mathbf{e}_{\mathbf{r}} + \left(\frac{10\,\mathrm{csgn}(1, \sin(t))\cos(5t)}{-3 + \cos(10t)} - \frac{50\,\mathrm{csgn}(\sin(t))\sin(5t)}{-3 + \cos(10t)} + \frac{100\,\mathrm{csgn}(\sin(t))\cos(5t)\sin(10t)}{(-3 + \cos(10t))^2} \right) \mathbf{e}_{\phi}$$

```
> combine(simplify(a));
```

$$(47 - 47\cos(2t) - 156\cos(10t) - 9\cos(22t) + 93\cos(12t) - 4\cos(18t) + 63\cos(8t) + 13\cos(20t)) / (-3\sqrt{12 - 4\cos(10t) - 12\cos(2t) + 2\cos(8t) + 2\cos(12t)} + \sqrt{12 - 4\cos(10t) - 12\cos(2t) + 2\cos(8t) + 2\cos(12t)}\cos(10t))\,\mathbf{e}_{\mathbf{r}} + (-50\,\mathrm{csgn}(1, \sin(t))\cos(5t) + 10\,\mathrm{csgn}(1, \sin(t))\cos(15t) + 450\,\mathrm{csgn}(\sin(t))\sin(5t) + 50\,\mathrm{csgn}(\sin(t))\sin(15t)) / (19 - 12\cos(10t) + \cos(20t))\,\mathbf{e}_{\phi}$$

Long term behaviour [would have been easy if the coords were (1,t,5*t) !]. In this case, the limit as t->infinity is a real mess.

```
>
```

Q3

```
> restart; with(VectorCalculus):
```

Some example. We can take our point b to be fixed at the origin, and the 'center' of the object to be moving. For simplicity, let the angular velocity be constant:

```
> omega := <4,5,6>;
```

$$\omega := 4\,\mathbf{e}_{\mathbf{x}} + 5\,\mathbf{e}_{\mathbf{y}} + 6\,\mathbf{e}_{\mathbf{z}}$$

Then we need to have

```
> Vcen = - CrossProduct(omega, <r1,r2,r3>);
```

$$Vcen = (-5\,r3 + 6\,r2)\,\mathbf{e}_{\mathbf{x}} + (-6\,r1 + 4\,r3)\,\mathbf{e}_{\mathbf{y}} + (-4\,r2 + 5\,r1)\,\mathbf{e}_{\mathbf{z}}$$

We can actually pick r to be constant, since r is a relative difference. For example

```
> Vcen = - CrossProduct(omega, <1,1,1>);
```

$$Vcen = \mathbf{e}_{\mathbf{x}} - 2\,\mathbf{e}_{\mathbf{y}} + \mathbf{e}_{\mathbf{z}}$$

In other words, from the center's perspective, it is always going in a particular direction -- but that is an illusion caused by the rotation of the frame.

A completely similar computation can be done for the acceleration.

A better example would be one where the situation is 'physical'.

```
>
```

Q4

Answer to this one to be provided later.

```
>
```

Q5

The goal is to solve Skew(omega) = diff(R(t),t)*Transpose(R(t)). We have that

```
> r := expand( <(1-t^2)*cos(Pi*t), (1-t^2)*sin(Pi*t),
  t^2>/sqrt((1-t^2)^2+t^4) );
```

$$r := \frac{(1-t^2)\cos(\pi t)}{\sqrt{1-2t^2+2t^4}}\mathbf{e}_{\mathbf{x}} + \frac{(1-t^2)\sin(\pi t)}{\sqrt{1-2t^2+2t^4}}\mathbf{e}_{\mathbf{y}} + \frac{t^2}{\sqrt{1-2t^2+2t^4}}\mathbf{e}_{\mathbf{z}}$$

```
> assume(t::RealRange(Open(-1),Open(1)));
```

One way to do this is to derive a system of axes. But if we look, r is already normalized:

```
> rn := Normalize(r);
```

$$rn := \frac{(1-t\sim^2)\cos(\pi t\sim)}{\sqrt{1-2t\sim^2+2t\sim^4}}\mathbf{e}_{\mathbf{x}} + \frac{(1-t\sim^2)\sin(\pi t\sim)}{\sqrt{1-2t\sim^2+2t\sim^4}}\mathbf{e}_{\mathbf{y}} + \frac{t\sim^2}{\sqrt{1-2t\sim^2+2t\sim^4}}\mathbf{e}_{\mathbf{z}}$$

So we can 'guess' that in spherical coordinates, this will be nice. In fact:

```
> rn := simplify(MapToBasis(Vector([1, arctan((1-t^2)/t^2),
  Pi*t],coords='spherical'), cartesian));
```

$$rn := -\frac{(-1+t\sim^2)\cos(\pi t\sim)}{\sqrt{1-2t\sim^2+2t\sim^4}}\mathbf{e}_{\mathbf{x}} - \frac{(-1+t\sim^2)\sin(\pi t\sim)}{\sqrt{1-2t\sim^2+2t\sim^4}}\mathbf{e}_{\mathbf{y}} + \frac{t\sim^2}{\sqrt{1-2t\sim^2+2t\sim^4}}\mathbf{e}_{\mathbf{z}}$$

Instead of bashing our heads against giant formulas for R and R', let's use this. We have that

```
> (rho, theta, phi) := (1, arctan((1-t^2)/t^2), Pi*t);
```

$$\rho, \theta, \phi := 1, \arctan\left(\frac{1-t\sim^2}{t\sim^2}\right), \pi t\sim$$

So we can simplify look at the motion matrix (2.31) to get that

```
> eqns := simplify({(-u3 = diff(theta,t)*cos(phi), u2 =
  -diff(theta,t)*sin(phi), -u1 = diff(phi,t))});
```

$$eqns := \left\{-u3 = -\frac{2\,t\sim\cos(\pi t\sim)}{1-2t\sim^2+2t\sim^4}, u2 = \frac{2\,t\sim\sin(\pi t\sim)}{1-2t\sim^2+2t\sim^4}, -u1 = \pi\right\}$$

```
> solve(eqns, {u1,u2,u3});
```

$$\left\{u2 = \frac{2\,t\sim\sin(\pi t\sim)}{1-2t\sim^2+2t\sim^4}, u3 = \frac{2\,t\sim\cos(\pi t\sim)}{1-2t\sim^2+2t\sim^4}, u1 = -\pi\right\}$$

```
>
```

Q6

```
> restart;
> x := <r*cos(omega*t), r*sin(omega*t), 1>:  F1 := <1,1,1>:
> W1 := int(F1.VectorCalculus[diff](x,t),t=0..2*Pi/omega);
```

$$W1 := 0$$

Work is 0 over that time period - force is conservative. In general,

```
> W := int(F1.VectorCalculus[diff](x,t),t=0..T) assuming T>0;
```

$$W := -r + r\cos(\omega T) + r\sin(\omega T)$$

```
> res := solve(diff(W,T)=0,T);
```

$$res := \frac{\pi}{4\,\omega}$$

```
> eval(diff(W,T,T),T=res);
```

$$-r\sqrt{2}\,\omega^2$$

So the work is a maximum when T = Pi/(4*omega) since the above is negative.

```
> F2 := <t,t,1>:
> W2 :=
  int(LinearAlgebra[DotProduct](F2,VectorCalculus[diff](x,t),
  conjugate=false),t=0..2*Pi/omega);
```

$$W2 := \frac{2\,r\,\pi}{\omega}$$

Work is not zero. In general:

```
> W :=
  int(LinearAlgebra[DotProduct](F2,VectorCalculus[diff](x,t),
  conjugate=false),t=0..T) assuming T>0;
```

$$W := \frac{r\,(-1 - \sin(\omega T) + \cos(\omega T)\,\omega T + \cos(\omega T) + \sin(\omega T)\,\omega T)}{\omega}$$

```
> solve(normal(diff(W,T))=0,T);
```

$$0, \frac{\pi}{4\,\omega}$$

There are maximums at 0 and at Pi/(4*omega).

```
> eval(diff(W,T,T),T=0), eval(diff(W,T,T),T=Pi/(4*omega));
```

$$r\,\omega, -\frac{r\sqrt{2}\,\omega\,\pi}{4}$$

The T=0 is a minimum, T=Pi/(4*omega) is a maximum.

```
>
```